
帰納的な定義と型付け
1 前回の課題
Lemma plusn0 n : n + 0 = n.

Proof. elim: n => /=. done. move=> n ->. done. Qed.

Lemma plusC m n : m + n = n + m.

Proof. elim: m => /= [|m ->]; by rewrite (plusn0,plusnS). Qed.

Lemma plusA m n p : m + (n + p) = (m + n) + p.

Proof. by elim: m => /= [|m ->]. Qed.

Lemma multn0 n : n * 0 = 0.

Proof. elim: n => //. Qed.

Lemma multC m n : m * n = n * m.

Proof. elim: m => /= [|m ->]; by rewrite (multn0,multnS). Qed.

Lemma multnDr m n p : (m + n) * p = m * p + n * p.

Proof. elim: m => /= [|m ->] //. by rewrite plusA. Qed.

Lemma multA m n p : m * (n * p) = (m * n) * p.

Proof. elim: m => /= [|m ->] //. by rewrite multnDr. Qed.

Lemma double_sum n : 2 * sum n = n * (n + 1).

Proof.

elim: n => /= [|n IH] //.

rewrite multnS -IH /=.

rewrite !plusnS !plusA /=.

rewrite ![_+0]plusC [_+n]plusC /=.

by rewrite !plusA.

Qed.

Lemma square_eq a b : (a + b) * (a + b) = a * a + 2 * a * b + b * b.

Proof.

rewrite multnDr.

rewrite (multC a) (multC b) !multnDr.

rewrite (multC b) /=.

rewrite [_+0]plusC /=.

by rewrite !plusA.

Qed.

Require Import ArithRing. (* nat における半環構造 *)

Lemma square_eq a b : (a + b) * (a + b) = a * a + 2 * a * b + b * b.

Proof. ring. Qed.

2 プログラムの型付け
型 τ, θ ::= nat | Z | . . . | θ → τ | τ × θ | ∀x:θ, τ データ型, 関数型, 直積, 依存積

型判定 Γ ⊢ M : τ Γ = x1 : τ1, . . . , xn : τn という仮定のもとで，M が型 τ をもつ．

型付け規則 Coqの式は以下の型付け規則によって型付けされる．

変数 Γ ⊢ x : τ (x : τはΓに含まれる)

抽象 Γ, x : θ ⊢ M : τ

Γ ⊢ (fun x:θ ⇒ M) : θ → τ

適用 Γ ⊢ M : θ → τ Γ ⊢ N : θ

Γ ⊢ M N : τ

定義 Γ ⊢ M : θ Γ, x : θ ⊢ N : τ

Γ ⊢ (let x := M in N) : τ

不動点 Γ, f : θ → τ, x : θ ⊢ M : τ

Γ ⊢ (fix f (x:θ) := M) : θ → τ

直積　 Γ ⊢ M : τ Γ ⊢ N : θ

Γ ⊢ (M,N) : τ × θ

射影 Γ ⊢ fst : τ × θ → τ Γ ⊢ snd : τ × θ → θ

抽象*
Γ, x : θ ⊢ M : τ

Γ ⊢ (fun x:θ ⇒ M) : ∀x:θ, τ
適用*

Γ ⊢ M : ∀x:θ, τ Γ ⊢ N : θ

Γ ⊢ M N : [N/x]τ

x ̸∈ fv(τ)のときに θ → τ = ∀x : θ, τ とおくと，抽象と適用が抽象*と適用* の特殊な場合に
なる．

型付けの例

Γ, x : nat ⊢ S : nat → nat Γ, x : nat ⊢ x : nat
Γ, x : nat ⊢ S x : nat

Γ ⊢ (fun x:nat ⇒ S x) : nat → nat
抽象

適用
Γ ⊢ O : nat

Γ ⊢ (fun x:nat ⇒ S x) O : nat
適用

3 命題と型の対応

カリー・ハワード同型により，命題論理と型理論 (型付λ計算)が対応している．
具体的には，以下のような対応が見られる．

命題 (論理式) 型
証明 (導出) プログラム
仮定∆ 型環境 Γ

⊃ →
∧ ∗

導出規則と型付け規則も基本的には 1対 1で対応している．それぞれの体系を少し修正すると
以下の定理がなりたつ．

定理 1 (Curry-Howard同型) ある同型 ⟨ ⟩ :命題→型が存在し，任意の∆と P について，導
出 Πより∆ ⊢ P が示せるならば，ΠからプログラムM が作れ，⟨∆⟩ ⊢ M : ⟨P ⟩．また，任意の
Γ,M, τ について型理論で Γ ⊢ M : τ が導出できれば，命題論理において ⟨Γ⟩−1 ⊢ ⟨τ⟩−1が導出で
きる．

修正の内容は二種類ある．
まず，上の不動点の規則は矛盾を生んでしまう．具体的には，θ = Trueと τ = Falseにすると，

以下の導出が可能になる．
Γ , f : True → False, x : True ⊢ f x : False

Γ ⊢ (fix f (x :θ) := f x ) : True → False

しかし，Coqの本当の不動点の規則はさらに f が xより小さな引数に適用されることを求めてい
るので，この矛盾が実際には起きない．本当の規則が複雑なのでここには書かない．
もう一つは，背理法に対する規則は Coqの型体系にはない．それは Coqは直観主義論理に基

いているからである．メリットとして，全ての証明が計算的な意味を持つ—証明は関数である．

4 帰納的な定義
Coqの帰納的データ型
前回は自然数の定義を見た．

Inductive nat : Set := O : nat | S : nat -> nat.

実は，Coqの全てのデータは帰納的データ型として定義される．1

Inductive prod (A B : Set) : Set := pair : A -> B -> prod A B.

Inductive sum (A B : Set) : Set := inl : A -> sum A B | inr : B -> sum A B.

帰納的データ型の値を作るのは構成を適用するだけでいい．しかし，分解するのに OCamlと
同様にパターンマッチングを使わなければならない．その型付け規則が複雑になる．以下のよう
なデータ型を考える．

Inductive t(a1 . . . an : Set) : Set :=

| c1 : τ11 → . . . → τ1k1 → t a1 . . . an

. . .

| cm : τm1 → . . . → τmkm → t a1 . . . an.

マッチング
Γ ⊢ M : t b1 . . . bn

Γ, xi1 : τi1[b1/a1, . . . , bn/an], . . . , xiki : τiki [. . .] ⊢ Mi : τ [ci xi1...xiki/x] (1 ≤ i ≤ m)

Γ ⊢ match M as x return τ with c1 x11 . . . x1k1 ⇒ M1 | . . . | cm xm1 . . . xmkm ⇒ Mm end : τ [M/x]

asと returnによって，返り値の型の中に入力を含めることができ，場合によって型が違うよう
な関数が作れる．それを手でやるのは難しいが，作戦 caseはこのパターンマッチングを構築し
てくれる．
帰納的データ型を定義すると，帰納法のための補題が自動的に定義されるが，定義は match

を使う．定義が再帰的でないとき，パターンマッチングだけで済む．再帰的なデータ型について
Fixpointが使われる．

1実際の定義を見ると，Setではなく Typeになっている．Typeは Setより一般的なので，Setとして使うことが
できる．さらに，prod A B は　 A*Bとして表示され，pair a bは (a,b)として表示される．Coqの Notationとい
う機能によって，帰納的データ型の表示方法を変えることができる．

Definition prod_ind (A B:Set) (P:prod A B -> Prop) :=

fun (f : forall a b, P (pair a b)) =>

fun p => match p as x return P x with pair a b => f a b end.

Check prod_ind.

: ∀ (A B : Set) (P : A * B -> Prop),

(∀ (a : A) (b : B), P (a, b)) -> ∀ p : A * B, P p

Definition sum_ind (A B:Set) (P:sum A B -> Prop) :=

fun (fl : forall a, P (inl _ a)) (fr : forall b, P (inr _ b)) =>

fun p => match p as x return P x

with inl a => fl a | inr b => fr b end.

Check sum_ind.

: ∀ (A B : Set) (P : A + B -> Prop),

(∀ a : A, P (inl B a)) -> (∀ b : B, P (inr A b)) ->

∀ p : A + B, P p

Fixpoint nat_ind (P:nat -> Prop) (f0:P O) (fn:forall n, P n -> P (S n))

(n : nat) {struct n} :=

match n as x return P x

with O => f0 | S m => fn m (nat_ind P f0 fn m) end.

Check nat_ind.

: ∀ P : nat -> Prop, P 0 -> (∀ n : nat, P n -> P (S n)) ->

∀ n : nat, P n

caseと elimはよく似ているが，前者が単なる場合分けを行うのに対して，後者が生成された
補題を利用しているので，効果が違ったりする．

Lemma plusn0 n : n + 0 = n.

Proof.

case: n.

- done.

(* ∀ n : nat, S n + 0 = S n *)

Restart.

move: n.

apply: nat_ind. (* elim の意味 *)

- done.

(* ∀ n : nat, n + 0 = n -> S n + 0 = S n *)

- move=> n /= -> //.

Qed.

リスト
リストは関数型プログラミングで最も使われているデータ構造である．

Require Import List. (* リストの関数と補題を使う *)

Print list.

Inductive list (A : Type) : Type :=

| nil : list A

| cons : A -> list A -> list A.

Check (cons 1 (cons 2 nil)).

1 :: 2 :: nil

: list nat

Import ListNotations. (* リストの記法を使う *)

Check [1; 2; 3].

[1; 2; 3]

: list nat

cons は二項演算子 :: として書ける．また，[a; b; c] という形でリスト cons a (cons b

(cons c nil))を書くこともできる．
リストの結合は appまたは++と書く．長さは length

Compute [1;2] ++ [3;4].

= [1; 2; 3; 4]

: list nat

Compute length [1;2;3].

= 3

: nat

それぞれ，Fixpointと場合分けで定義されている．
Print app.

fix app (l m : list A) struct l : list A :=

match l with

| [] => m

| a :: l1 => a :: app l1 m

end

: forall A : Type, list A -> list A -> list A

Print length.

fix length (l : list A) : nat :=

match l with

| [] => 0

| _ :: l’ => S (length l’)

end

: forall A : Type, list A -> nat

自然数とほぼ同じ形で帰納法が使える．
Lemma length_app A (l1 l2 : list A) :

length (l1 ++ l2) = length l1 + length l2.

Proof.

elim: l1 => /=. done.

move=> a l1 IH. (* 自然数と違い，リストは中身がある *)

by rewrite IH.

Qed.

多項式の表現と計算 多項式をリストとして書く場合には，係数を字数の上る順に書く．x3+3x+2

は [2; 3; 0; 1]になる．
計算はHornerの公式が最も簡単．
Fixpoint horner (l : list nat) x :=

match l with

| nil => 0

| a :: l’ => a + x * horner l’ x

end.

Compute horner [2; 3; 0; 1] 4.

= 78

: nat

練習問題 4.1 先週証明した補題を使って，Horner の公式が以下の多項式の定義と同値であるこ
とを示しなさい．ただし，関数 exp x n = xnは Fixpointを使って定義しなさい．
Fixpoint exp (x n : nat) := 1. (* x の n 乗 *)

Fixpoint eval_poly n (l : list nat) x :=

match l with

| nil => 0

| a :: l’ => a * exp x n + eval_poly (S n) l’ x

end.

Lemma eval_polyS n l x : eval_poly (S n) l x = x * eval_poly n l x.

Proof. elim: l n. Admitted.

Lemma horner_ok l x : horner l x = eval_poly 0 l x.

Proof. suff : forall n, exp x n * horner l x = eval_poly n l x. Admitted.

帰納的述語
Coqでは帰納的な定義は SetだけでなくPropでもできる．この場合，パラメータは場合によっ

て変わることが多い．

Inductive t : τ1 → . . . → τn → Prop :=

| c1 : ∀(x1 : τ11) . . . (xk1 : τ1k1), t θ11 . . . θ1n

. . .

| cm : ∀(x1 : τm1) . . . (xk1 : τmkm), t θm1 . . . θmn

マッチング
Γ ⊢ M : t b1 . . . bn Γ, xi1 : τi1, . . . , xiki : τiki ⊢ Mi : τ [θi1 . . . θin/x1 . . . xn] (1 ≤ i ≤ m)

Γ ⊢ match M in t x1 . . . xn return τ with c1 x11 . . . x1k1 ⇒ M1 | . . . | cm xm1 . . . xmkm ⇒ Mm end

: τ [b1, . . . , bn/x1 . . . xn]

(* 偶数の定義 *)

Inductive even : nat -> Prop :=

| even_O : even O

| even_SS : forall n, even n -> even (S (S n)).

(* 帰納的述語を証明する定理 *)

Theorem even_double n : even (n + n).

Proof.

elim: n => /= [|n IH].

- apply: even_O.

- rewrite -plus_n_Sm.

by apply: even_SS.

Qed.

(* 帰納的述語に対する帰納法もできる *)

Theorem even_plus m n : even m -> even n -> even (m + n).

Proof.

elim: m => //=.

Restart.

move=> Hm Hn.

elim: m / Hm => //= m IH.

apply: even_SS.

Qed.

(* 矛盾を導き出す *)

Theorem one_not_even : ~ even 1.

Proof.

case.

Restart.

move=> He.

inversion He.

Show Proof. (* 証明が複雑で、SSReflectでは様々な理由で避ける *)

Restart.

move H: 1 => one He. (* move H: exp => pat は H: exp = patを作る *)

by case: He H.

Qed.

(* 等式を導き出す *)

Theorem eq_pred m n : S m = S n -> m = n.

Proof.

case. (* 等式を分解する *)

done.

Qed.

実は Coqの論理結合子のほとんどが帰納的述語として定義されている．

Inductive and (A B : Prop) : Prop := conj : A -> B -> A /\ B.

Inductive or (A B : Prop) : Prop :=

or_introl : A -> A \/ B | or_intror : B -> A \/ B.

Inductive ex (A : Type) (P : A -> Prop) : Prop :=

ex_intro : forall x : A, P x -> exists x, P x.

Inductive False : Prop := .

and，orや exについて caseが使えた理由がこの定義方法である．
しかも，Falseは最初からあるものではなく，構成子のない述語として定義されている．生成

される帰納法の補題をみると面白い．
Print False_ind.

fun (P : Prop) (f : False) => match f return P with end

: ∀ P : Prop, False -> P

ちょうど，矛盾の規則に対応している．作戦　 elimでそれが使える．
Theorem contradict (P Q : Prop) : P -> ~P -> Q.

Proof. move=> p. elim. exact: p. Qed.

練習問題 4.2 以下の定理を証明しなさい．
Module Odd.

Inductive odd : nat -> Prop :=

| odd_1 : odd 1

| odd_SS : forall n, odd n -> odd (S (S n)).

Theorem even_odd n : even n -> odd (S n). Abort.

Theorem odd_even n : odd n -> even (S n). Abort.

Theorem even_not_odd n : even n -> ~odd n. Abort.

End Odd.

